Un exemple complet de diagonalisation en Maple

PC*2

20 novembre 2001

Table des matières

1 Les commandes Maple

– Initialiser la bibliothèque d'algèbre linéaire :

$$with(linalg)$$

Si on met un " ; ", on obtient la liste des fonctions de la bibliothèque.

– Créer une matrice : On se donne la liste des vecteurs lignes qui sont eux mêmes des listes :

$$A := matrix([L_1, L_2, \dots, L_n]);$$

$L_i = [a_{i,1}, a_{i,2}, \dots, a_{i,p}]$.
Créer la matrice diagonale $diag(a_1, \dots, a_n)$

$$diag(a_1, \dots, a_n)$$

(En particulier l'identité)

– Obtenir une matrice A :

$$evalm(A)$$

L'invocation de A ne retourne que le nom de la matrice.

– Accéder à l'élément $a_{i,j}$ de A :

$$A[i, j]$$

– Pivoter les lignes :

$$pivot(A, i, j)$$

En version 3 : Exécute l'opération :

$$\text{de } k = i + 1 \text{ à } n \ L_k \leftarrow L_k - \frac{a_{k,j}}{a_{i,j}} L_i$$

En version 4 : Même chose mais pour $k \neq i$.
Il ya déclenchement d'une erreur si $a_{i,j} = 0$.

– Permuter deux lignes :

$$swaprow(A, i, j)$$

swapcol pour les colonnes

– Faire simplement l'opération :

$$L_j \leftarrow L_j + \lambda L_i$$

$$addrow(A, i, j, lambda)$$

addcol pour les colonnes

– Base du noyau de l'application linéaire canoniquement associée à A [1]

$$nullspace(A)$$

– Liste des valeurs propres avec leur multiplicité respective et une base du sous espace propre associé :

$$eigenvects(A)$$

– Polynôme caractéristique :

$$charpoly$$

Déterminant : *det*

[1]ie représentée par A lorsque les espaces de départ et d'arrivée sont des $\mathbf{R}^n$ rapportés à leur base canonique respective

- Somme de deux matrices A et B :

$$A + B \text{ ou } add(A, B)$$

Idem pour $A - B$.
- Produit par un scalaire λ :

$$\lambda * A$$

- produit de A et B :

$$A\& * B$$

- Inverse de A :

$$A^{(-1)}$$

- Transposée de A :

$$transpose(A)$$

- Substituer a à x dans la matrice A :

$$subs(x = a, evalm(A))$$

- Appliquer la fonction f à tous les éléments de A :

$$map(f, evalm(A))$$

- Quotient de la division euclidienne du polynôme P par le polynôme Q par rapport à la variable x :

$$quo(P, Q, x)$$

$\deg(P - UQ) < \deg(P)$ *si* U *est ledit quotient*
- Union des ensembles E et F (entre accolades) :

$$E \ union \ F$$

- Conversion d'un ensemble S en liste :

$$convert(S, list)$$

La fonction convert prend de nombreuses formes. Regarder la rubrique d'aide.

2 Étude d'un exemple

```
with(linalg):
A:=matrix([[a^2,a*b,a*b,b^2],
[a*b,a^2,b^2,a*b],[a*b,b^2,a^2,a*b],
[b^2,a*b,a*b,a^2]]);
I4:=diag(1,1,1,1);
M:=evalm(A-x*I4);
```

$$A = \begin{bmatrix} a^2 & ab & ab & b^2 \\ ab & a^2 & b^2 & ab \\ ab & b^2 & a^2 & ab \\ b^2 & ab & ab & a^2 \end{bmatrix} \quad I4 = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$M = \begin{bmatrix} a^2 - x & ab & ab & b^2 \\ ab & a^2 - x & b^2 & ab \\ ab & b^2 & a^2 - x & ab \\ b^2 & ab & ab & a^2 - x \end{bmatrix}$$

On règle le pivot de la première colonne de façon à faire apparaître le moins de polynômes possibles dans la suite :

```
M1:=swaprow(M,1,4);
M2:=pivot(M1,1,1);
```

$$M1 = \begin{bmatrix} b^2 & ab & ab & a^2 - x \\ ab & a^2 - x & b^2 & ab \\ ab & b^2 & a^2 - x & ab \\ a^2 - x & ab & ab & b^2 \end{bmatrix}$$

$$M2 = \left[\begin{array}{cccc} b^2 & ab & ab & a^2 - x \\ 0 & -x & -a^2 + b^2 & \frac{a\left(-a^2+x+b^2\right)}{b} \\ 0 & -a^2 + b^2 & -x & \frac{a\left(-a^2+x+b^2\right)}{b} \\ 0 & \frac{a\left(-a^2+x+b^2\right)}{b} & \frac{a\left(-a^2+x+b^2\right)}{b} & -\frac{a^4-2\,a^2x+x^2-b^4}{b^2} \end{array}\right]$$

On s'arrange pour prendre le pivot constant :

```
M3:=swaprow(M2,2,3);
M4:=pivot(M3,2,2);
```

$$M3 = \left[\begin{array}{cccc} b^2 & ab & ab & a^2 - x \\ 0 & -a^2 + b^2 & -x & \frac{a\left(-a^2+x+b^2\right)}{b} \\ 0 & -x & -a^2 + b^2 & \frac{a\left(-a^2+x+b^2\right)}{b} \\ 0 & \frac{a\left(-a^2+x+b^2\right)}{b} & \frac{a\left(-a^2+x+b^2\right)}{b} & -\frac{a^4-2\,a^2x+x^2-b^4}{b^2} \end{array}\right]$$

$$M4 = \left[\begin{array}{cccc} b^2 & 0 & -\frac{ab\left(-a^2+x+b^2\right)}{a^2-b^2} & \frac{xb^2}{a^2-b^2} \\ 0 & -a^2 + b^2 & -x & \frac{a\left(-a^2+x+b^2\right)}{b} \\ 0 & 0 & \frac{x^2-a^4+2\,a^2b^2-b^4}{a^2-b^2} & -\frac{a\left(-a^2+x+b^2\right)^2}{(a^2-b^2)b} \\ 0 & 0 & -\frac{a\left(-a^2+x+b^2\right)^2}{(a^2-b^2)b} & \frac{x^2-a^4+2\,a^2b^2-b^4}{a^2-b^2} \end{array}\right]$$

Tous les pivots potentiels sont des polynômes en x. On fait baisser le degré via une division de polynômes.

```
qu:=quo(M4[3,3],M4[4,3],x);
```

$$-\frac{b}{a}$$


```
M5:=map(simplify,evalm(addrow(M4,4,3,-qu)));
qu:=quo(M5[4,3],M5[3,3],x);
M6:=map(factor,evalm(map(simplify,evalm(addrow(M5,3,4,-qu)))));
```

$$M5 = \left[\begin{array}{cccc} b^2 & 0 & -\frac{ab\left(-a^2+x+b^2\right)}{a^2-b^2} & \frac{xb^2}{a^2-b^2} \\ 0 & -a^2 + b^2 & -x & \frac{a\left(-a^2+x+b^2\right)}{b} \\ 0 & 0 & -2\,a^2 + 2\,b^2 + 2\,x & -\frac{a^4-2\,a^2x+x^2-b^4}{ab} \\ 0 & 0 & -\frac{a\left(-a^2+x+b^2\right)^2}{(a^2-b^2)b} & \frac{x^2-a^4+2\,a^2b^2-b^4}{a^2-b^2} \end{array}\right]$$

$$qu = \frac{a}{2\,b} - \frac{ax}{(2\,a^2 - 2\,b^2)\,b}$$

$$M6 = \left[\begin{array}{cccc} b^2 & 0 & -\frac{ab\left(-a^2+x+b^2\right)}{(a-b)(a+b)} & \frac{b^2x}{(a-b)(a+b)} \\ 0 & -\,(a-b)\,(a+b) & -x & \frac{a\left(-a^2+x+b^2\right)}{b} \\ 0 & 0 & -2\,a^2 + 2\,b^2 + 2\,x & -\frac{\left(-a^2+x+b^2\right)\left(x-b^2-a^2\right)}{ab} \\ 0 & 0 & 0 & -\frac{\left(x-a^2-2\,ab-b^2\right)\left(-a^2+x+b^2\right)\left(x-a^2+2\,ab-b^2\right)}{2\,b^2(a-b)(a+b)} \end{array}\right]$$

```
V1:=nullspace(subs(x=a^2-b^2,evalm(M6)));
V2:=nullspace(subs(x=a^2+2*a*b+b^2,evalm(M6)));
V3:=nullspace(subs(x=a^2-2*a*b+b^2,evalm(M6)));
P:=transpose(matrix(convert(V1 union V2 union V3,list)));
```

$$V1 = \{[0, -1, 1, 0], [-1, 0, 0, 1]\}$$

$$V2 = \{[1, 1, 1, 1]\}$$

$$V3 = \{[1, -1, -1, 1]\}$$

$$P = \left[\begin{array}{cccc} 1 & 0 & -1 & 1 \\ 1 & -1 & 0 & -1 \\ 1 & 1 & 0 & -1 \\ 1 & 0 & 1 & 1 \end{array}\right]$$

```
Dia:=evalm(P^(-1)&*A&*P);
```

$$Dia = \begin{bmatrix} a^2 + 2\,ab + b^2 & 0 & 0 & 0 \\ 0 & a^2 - b^2 & 0 & 0 \\ 0 & 0 & a^2 - b^2 & 0 \\ 0 & 0 & 0 & a^2 - 2\,ab + b^2 \end{bmatrix}$$